
** Eileen Braman (TESS 751)
** Project title: Exploring the Role of Constitutional Considerations, Public
**		Support and Personal Preferences in Citizen Assessments of Proposed
**		Congressional Action on Immigration and Gun Control

clear all
use braman751.dta, clear


/*

Stated-Hyp1: "Policy views should predominate in judgments of desirability and 
information about compliance with rules should be significant in judgments 
of the legitimacy of proposed action."

	Test-Hyp1: Policy views should predict judgments of desirability of proposed
	laws.
		** Note: Ideally this would be that Policy views should be more 
		important than compliance with rules for judgments about the 
		desirability of laws.  But not sure how to compare this given that 
		views are measured on a 7-point scale and compliance with rules is
		a 3 part trichotomy, so decided not to focus on word "predominate" and
		instead make parallel to Test-Hyp2
	
	Test-Hyp2: Compliance with rules should be significant for judgments of
	legitimacy of a proposed action.

Stated-Hyp2: "One political factor that could influence perceptions of legitimacy is 
the level of popular support for proposed action."

	Test-Hyp3: Average legitimacy ratings will be higher in the 85 percent
	condition than the 15 percent condition.

Stated-Hyp3: "People may be more likely to believe laws that they agree with 
are the result of legitimate government authority."

	Test-Hyp4: Agreement with the proposed law will be positively associated
	with legitimacy.

Stated-Hyp4: "Those who support a strong role for the federal government may be more
likely to see Congressional action as appropriate where it is in conflict
with state authority."

	* note: from context of proposal text clear that "appropriate" is "legitimacy"

	Test-Hyp5: Support for a strong role for federal government will be
	associated with seeing bill as legitimate 

Stated-Hyp5: "The political variables identified will play more of a role 
where constitutional cues regarding authority are unclear."

	Test-Hyp6: Relationship between public support and legitimacy assessment
	will be higher in the ynclear condition than the clear/clearly not conditions.
	
	Test-Hyp7: Relationship between agreement with proposed law and legitimacy
	assessment will be higher in the unclear condition than the clear/clearly 
	not conditions.

	Test-Hyp8: Relationship between support for federalism and assessment
	of legitimacy will be stronger in the unclear condition than the clear/
	clearly not conditions.
	
*/

** MEASURES OF EXPERIMENTAL CONDITIONS

/*

Conditions are defined in raw data by a single variable XTESS225, coded as follows:

      1 Gun Control Scenario, Clear Authority, 85% of public supports proposed legislation
      2 Gun Control Scenario, Clear Authority, 15% of public supports proposed legislation
      3 Gun Control Scenario, Unclear Authority, 85% of public supports proposed legislation
      4 Gun Control Scenario, Unclear Authority, 15% of public supports proposed legislation
      5 Gun Control Scenario, Clearly No Authority, 85% of public supports proposed legislation
      6 Gun Control Scenario; Clearly No Authority; 15% of public supports proposed legislation
      7 Immigration Scenario, Clear Authority, 85% of public supports proposed legislation
      8 Immigration Scenario, Clear Authority, 15% of public supports proposed legislation
      9 Immigration Scenario, Unclear Authority, 85% of public supports proposed legislation
     10 Immigration Scenario, Unclear Authority, 15% of public supports proposed legislation
     11 Immigration Scenario, Clearly No Authority, 85% of public supports proposed legislation
     12 Immigration Scenario, Clearly No Authority, 15% of public supports proposed legislation

*/

* issue context -- gun control or immigration

recode XTESS225 (1/6=1) (7/12=0) (*=.), gen(gun_control)
recode XTESS225 (1/6=0) (7/12=1) (*=.), gen(immigration)

* expert assessment of constitutional authority

recode XTESS225 (1/2=1) (7/8=1) (3/4=2) (9/10=2) (5/6=3) (11/12=3) (*=.), gen(authority)
label define authority 1 "Clear" 2 "Unclear" 3 "Clearly no"
label values authority authority

	* clarity vs. not
	recode authority (1 3 = 1 "Clear") (2=0 "Unclear"), gen(authority_clear)

* level of public support for proposed action

recode XTESS225 (1 3 5 7 9 11 = 1) (2 4 6 8 10 12 = 0) (*=.), gen(support85)
recode XTESS225 (1 3 5 7 9 11 = 0) (2 4 6 8 10 12 = 1) (*=.), gen(support15)

tab gun_control authority
tab gun_control support85
tab authority support85

** OUTCOMES

* desirability

recode B10 (-1=.), gen(desirability)

* legitimacy

recode B12 (-1=.) (1=7) (2=6) (3=5) (4=4) (5=3) (6=2) (7=1) (*=.), gen(legitimacy)

** MODERATORS

* policy views

	* Note: coded so that higher values are consistent with the proposed legislation

	* gun control
	recode B05 (-1=.), gen(gun_control_views)

	* immigration
	recode B05 (-1=.), gen(immigration_views)

	* consolidated views variable
		* gun control views for people who got the gun control scenario, and
		* immigration views for people who got the immigration scenario

	gen views = gun_control_views if gun_control == 1	
	replace views = immigration_views if immigration == 1	

* support for federal authority

recode B07 (-1=.) (1=6) (2=5) (3=4) (4=3) (5=2) (6=1) (*=.), gen(federalist)

** ANALYSIS

/*
	Test-Hyp1: Policy views should be more important than compliance with rules
	for judgments about the desirability of laws.
	
	Note that there is not an ideal way to test this given that one is measured
	as a likert scale and the other as a trichotomy.  Comparing R2.
	
*/

regress desirability views 
// SUPPORTED at p .000
tess views +, init(Braman751)

/*
	Test-Hyp2: Compliance with rules should be significant for judgments of
	legitimacy of a proposed action.
*/

regress legitimacy i.authority if authority != 2
// SUPPORTED difference between clear and clearly no is p = .001
tess 3.authority

/*
	Test-Hyp3: Average legitimacy ratings will be higher in the 85 percent
	condition than the 15 percent condition.
*/

regress legitimacy i.support85 
// NOT SUPPORTED - p = .10
tess 1.support85

/*
	Test-Hyp4: Agreement with the proposed law will be positively associated
	with legitimacy.
*/

regress legitimacy views
// SUPPORTED p = .000
tess views +

/*
	Test-Hyp5: Support for a strong role for federal government will be
	associated with seeing bill as legitimate 
*/

regress legitimacy federalist
// SUPPORTED p = .000
tess federalist 

/*
	Test-Hyp6: Relationship between public support and legitimacy assessment
	will be higher in the unclear condition than the clear/clearly not conditions.
*/

	* Note: I conducted this as a single joint test rather than separate tests
	* vs. the clear/clearly not conditions. It turns out that one of the 
	* tests going into the joint test is significant and the other isn't, resulting
	* in the joint test being null, but my decision to the joint test was prior
	* to seeing that result.

	regress legitimacy i.support85##ib1.authority_clear
	// NOT SUPPORTED - p = .209
	tess 1.support85#0.authority_clear +

/*	
	Test-Hyp7: Relationship between agreement with proposed law and legitimacy
	assessment will be higher in the unclear condition than the clear/clearly 
	not conditions.
*/

	sum views
	gen viewscen = views-r(mean) // center variable for interaction terms

	regress legitimacy c.viewscen##ib1.authority_clear
	// NOT SUPPORTED - p = .007 (wrong direction)
	tess c.viewscen#0.authority_clear +
	
/*
	Test-Hyp8: Relationship between support for federalism and assessment
	of legitimacy will be stronger in the unclear condition than the clear/
	clearly not conditions.
*/

	sum federalist
	gen fedcen = federalist - r(mean) // center variable for interaction terms
	regress legitimacy c.fedcen##ib1.authority_clear
	// NOT SUPPORTED - p = 0.068 (wrong direction)
	tess c.fedcen#0.authority_clear +
